SRR

```
(* date : 2017 May 30 *)
(* author : Jinsu Kim modified by SCPARK*)
(* note :
 Primordial black holes in critical Higgs inflation. In this file,
all the parameters should be chosen at
 the top quark mass initially. For a parameter scan,
please refer to my another Mathematica file. One may use this file as
 a play ground. However, please do NOT make any change in this file.
*)
```

Primordial Black Holes in Critical Higgs Inflation
(* Mt = 170.850993 GeV *)

```
Quit[]
```

Initial conditions at $\mu = M_t$

```
(* PDG2016 & 1307.3536 *)
```

```
Mt = 170.850993; (*top quark mass (165-172) GeV*)
xihMt = 12.181; (*non-minimal coupling (0-400000)*)
Mp = 1.221 × 10^19;
MpR = 2.4 * 10^18; (*reduced Planck mass*)
MW = 80.385;
MZ = 91.1876;
MH = 125.09;
alphasMZ = 0.1182;
lHMt = 0.12604 + 0.00206 (MH - 125.15) - 0.00004 (Mt - 173.34);
ytMt = 0.93690 + 0.00556 (Mt - 173.34) -
    0.00003 (MH - 125.15) - 0.00042 * (alphasMZ - 0.1184) / 0.0007;
g3Mt = 1.1666 + 0.00314 * (alphasMZ - 0.1184) / 0.0007 - 0.00046 (Mt - 173.34);
g2Mt = 0.64779 + 0.00004 (Mt - 173.34) + 0.00011 * (MW - 80.384) / 0.014;
g1Mt = 0.35830 + 0.00011 (Mt - 173.34) - 0.00020 * (MW - 80.384) / 0.014;
(* mr c++ library: *)
(*g1Mt=0.4138291005958860`*)
(*g2Mt=0.648116*)
(*g3Mt=1.16528*)
(*ytMt=0.9352*)
(*lHMt=0.125886*)
```

RGE

Renormalisation scale

```
(*t[μ_]:=Log[μ/Mt]*)
```

$$h[t\_] := \left(yt[t] \Big/ \sqrt{2.}\right)^{-1} Mt * Exp[t]$$

Suppression factor

$$sh[t\_] := \frac{1 + xih[t] * h[t]^2 \Big/ MpR^2}{1 + \left(1 + 6\, xih[t]\right) xih[t] * h[t]^2 \Big/ MpR^2}$$

1-loop beta functions

```
betalH1[t_] := 6 (1 + 3 sh[t]^2) lH[t]^2 + 12 lH[t] yt[t]^2 -
    6 yt[t]^4 - 3 lH[t] (3 g2[t]^2 + g1[t]^2) + 3/8 (2 g2[t]^4 + (g1[t]^2 + g2[t]^2)^2);

betag11[t_] := (81 + sh[t])/12 g1[t]^3;
betag21[t_] := -(39 - sh[t])/12 g2[t]^3;
betag31[t_] := -7 g3[t]^3;

betayt1[t_] := yt[t] * ((23/6 + 2/3 sh[t]) yt[t]^2 - (8 g3[t]^2 + 9/4 g2[t]^2 + 17/12 g1[t]^2));
betaxih1[t_] := betaxih1[t] = 0;
(*betaxih1[t_]:=(6 (1+sh[t]) lH[t]+6yt[t]^2-3/2 (g1[t]^2+3g2[t]^2)) (xih[t]+1/6);*)

gamma1[t_] := -(9/4 g2[t]^2 + 3/4 g1[t]^2 - 3 yt[t]^2);
```

2-loop beta functions

```
betalH2[t_] :=
  1/48 ((912 + 3 sh[t]) g2[t]^6 - (290 - sh[t]) g1[t]^2 g2[t]^4 - (560 - sh[t]) g1[t]^4 g2[t]^2 -
       (380 - sh[t]) g1[t]^6) + (38 - 8 sh[t]) yt[t]^6 -
   yt[t]^4 (8/3 g1[t]^2 + 32 g3[t]^2 + (12 - 117 sh[t] + 108 sh[t]^2) lH[t]) +
   lH[t] * (-1/8 (181 + 54 sh[t] - 162 sh[t]^2) g2[t]^4 +
       1/4 (3 - 18 sh[t] + 54 sh[t]^2) g1[t]^2 g2[t]^2 + 1/24 (90 + 377 sh[t] + 162 sh[t]^2) g1[t]^4 +
       (27 + 54 sh[t] + 27 sh[t]^2) g2[t]^2 lH[t] + (9 + 18 sh[t] + 9 sh[t]^2) g1[t]^2 lH[t] -
       (48 + 288 sh[t] - 324 sh[t]^2 + 624 sh[t]^3 - 324 sh[t]^4) lH[t]^2) +
   yt[t]^2 (-9/4 g2[t]^4 + 21/2 g1[t]^2 g2[t]^2 - 19/4 g1[t]^4 +
       lH[t] * (45/2 g2[t]^2 + 85/6 g1[t]^2 + 80 g3[t]^2 - (36 + 108 sh[t]^2) lH[t])));

betag12[t_] := 199/18 g1[t]^5 + 9/2 g1[t]^3 g2[t]^2 + 44/3 g1[t]^3 g3[t]^2 - 17/6 sh[t] * g1[t]^3 yt[t]^2;
betag22[t_] := 3/2 g1[t]^2 g2[t]^3 + 35/6 g2[t]^5 + 12 g2[t]^3 g3[t]^2 - 3/2 sh[t] * g2[t]^3 yt[t]^2;
betag32[t_] := 11/6 g1[t]^2 g3[t]^3 + 9/2 g2[t]^2 g3[t]^3 - 26 g3[t]^5 - 2 sh[t] * g3[t]^3 yt[t]^2;

betayt2[t_] :=
  yt[t] * (-23/4 g2[t]^4 - 3/4 g1[t]^2 g2[t]^2 + 1187/216 g1[t]^4 + 9 g2[t]^2 g3[t]^2 + 19/9 g1[t]^2 g3[t]^2 -
      108 g3[t]^4 + (225/16 g2[t]^2 + 131/16 g1[t]^2 + 36 g3[t]^2) sh[t] * yt[t]^2 +
      6 (-2 sh[t]^2 yt[t]^4 - 2 sh[t]^3 yt[t]^2 lH[t] + sh[t]^2 lH[t]^2));

gamma2[t_] := -(271/32 g2[t]^4 - 9/16 g1[t]^2 g2[t]^2 - 431/96 sh[t] * g1[t]^4 -
      5/2 (9/4 g2[t]^2 + 17/12 g1[t]^2 + 8 g3[t]^2) yt[t]^2 + 27/4 sh[t] * yt[t]^4 - 6 sh[t]^3 lH[t]^2);
```

RGEs

```
betalH[t_] := 1/(16 π^2) betalH1[t] + 1/(16 π^2)^2 betalH2[t];
betag1[t_] := 1/(16 π^2) betag11[t] + 1/(16 π^2)^2 betag12[t];
betag2[t_] := 1/(16 π^2) betag21[t] + 1/(16 π^2)^2 betag22[t];
betag3[t_] := 1/(16 π^2) betag31[t] + 1/(16 π^2)^2 betag32[t];
betayt[t_] := 1/(16 π^2) betayt1[t] + 1/(16 π^2)^2 betayt2[t];
betaxih[t_] := 1/(16 π^2) betaxih1[t];
gamma[t_] := 1/(16 π^2) gamma1[t] + 1/(16 π^2)^2 gamma2[t];
```

Solving the RGEs

```
mustart = Mt;
muend = 6.5 * Mp;
tmin = Log[mustart / mustart]
tmax = Log[muend / mustart]
```

```
0.
```

```
40.6798
```

```
RGEsol = NDSolve[{
    g1'[t] == betag1[t] / (1 + gamma[t]),
    g2'[t] == betag2[t] / (1 + gamma[t]),
    g3'[t] == betag3[t] / (1 + gamma[t]),
    lH'[t] == betalH[t] / (1 + gamma[t]),
    yt'[t] == betayt[t] / (1 + gamma[t]),
    xih'[t] == betaxih[t],
    g1[tmin] == g1Mt,
    g2[tmin] == g2Mt,
    g3[tmin] == g3Mt,
    lH[tmin] == lHMt,
    yt[tmin] == ytMt,
    xih[tmin] == xihMt
   }, {g1, g2, g3, yt, lH, xih}, {t, tmin, tmax},
   AccuracyGoal → Automatic, PrecisionGoal → Automatic, MaxSteps → Infinity];
```

```
g1sol[t_] := g1[t] /. Flatten[RGEsol];
g2sol[t_] := g2[t] /. Flatten[RGEsol];
g3sol[t_] := g3[t] /. Flatten[RGEsol];
ytsol[t_] := yt[t] /. Flatten[RGEsol];
lHsol[t_] := lH[t] /. Flatten[RGEsol];
xihsol[t_] := xih[t] /. Flatten[RGEsol];
gammasol[t_] := gamma[t] /. Flatten[RGEsol];
hsol[t_] := h[t] /. Flatten[RGEsol];
ht[t_] := h'[t] /. Flatten[RGEsol];
lHdsol[t_] := lH'[t] /. Flatten[RGEsol];
lHddsol[t_] := lH''[t] /. Flatten[RGEsol];
httest[t_] := D[hsol[x], x] /. {x → t};

xihdsol[t_] := xih'[t] /. Flatten[RGEsol];

tcrit = FindRoot[lHdsol[t] == 0, {t, tmax}][[1, 2]]

35.3911
```

RG-improved effective potential

```
GHexact[t_?NumberQ] :=
  Exp[-NIntegrate[gammasol[x] / (gammasol[x] + 1), {x, 0, t}, AccuracyGoal → Automatic,
     PrecisionGoal → Automatic, Method → {Automatic, "SymbolicProcessing" → 0}]];

tstep = (tmax - tmin) / 100.;
GHlist = Table[{x, GHexact[t] /. {t → x}}, {x, tmin, tmax, tstep}];
GHintpln[t_] := Interpolation[GHlist][t];

VJordanexact[t_] := (1. / 4) * lHsol[t] * GHexact[t] ^ 4 * hsol[t] ^ 4;
VJordan[t_] := (1. / 4) * lHsol[t] * GHintpln[t] ^ 4 * hsol[t] ^ 4;

NMC[t_] := 1 + xihsol[t] GHintpln[t]^2 hsol[t]^2 / MpR^2;
NMCt[t_] := D[NMC[x], x] /. x → t;
NMCh[t_] := NMCt[t] / ht[t];
VE[t_] := VJordan[t] / NMC[t]^2;
VEt[t_] := D[VE[x], x] /. x → t;
VEtt[t_] := D[VEt[x], x] /. x → t;
VEh[t_] := VEt[t] / ht[t];
VEht[t_] := D[VEh[x], x] /. x → t;
VEhh[t_] := VEht[t] / ht[t];
```

```
dchidh[t_] := Sqrt[GHintpln[t]^2 / NMC[t] + 3 MpR^2/(2 NMC[t]^2) (NMCh[t])^2];
NMCtest[t_] := 1 + xihsol[t] 1^2 hsol[t]^2 / MpR^2;
NMCttest[t_] := D[NMCtest[x], x] /. x → t;
NMChtest[t_] := NMCttest[t] / ht[t];
dchidhtest[t_] := Sqrt[1^2 / NMCtest[t] + 3 MpR^2/(2 NMCtest[t]^2) (NMChtest[t])^2];

VEchi[t_] := VEh[t] / dchidh[t];
VEchit[t_] := D[VEchi[x], x] /. x → t;
VEhchi[t_] := VEchit[t] / ht[t];
VEchichi[t_] := VEhchi[t] / dchidh[t];
VEchichit[t_] := D[VEchichi[x], x] /. x → t;
VEchichih[t_] := VEchichit[t] / ht[t];
VEchichichi[t_] := VEchichih[t] / dchidh[t];
```

Cosmological observables

```
α = -0.7296; (* what is this??*)

ϵV[t_] := MpR^2 / 2 (VEchi[t] / VE[t])^2;
ηV[t_] := MpR^2 (VEchichi[t] / VE[t]);
ξV2[t_] := MpR^4 ((VEchi[t] VEchichichi[t]) / VE[t]^2);

Ps[t_] := 1 / (24 * Pi^2 * MpR^4) * VE[t] / ϵV[t];
ns[t_] := 1 - 6 ϵV[t] + 2 ηV[t] - 2/3 (5 + 36 α) ϵV[t]^2 +
   2 (-1 + 8 α) ϵV[t] ηV[t] + (2 ηV[t]^2)/3 + (2/3 - 2 α) ξV2[t];
r[t_] := 16 ϵV[t] - (64 ϵV[t]^2)/3 + 64 α ϵV[t]^2 + (32 ϵV[t] ηV[t])/3 - 32 α ϵV[t] ηV[t];
dnsdlnk[t_] := -24 ϵV[t]^2 + 16 ϵV[t] ϵV[t] - 2 ξV2[t];

Nef[tpivot_?NumberQ, tend_?NumberQ] :=
  NIntegrate[1 / MpR^2 (dchidh[x9]) (ht[x9]) (VE[x9] / VEchi[x9]),
   {x9, tend, tpivot}, AccuracyGoal → Automatic, PrecisionGoal → Automatic,
   Method → {Automatic, "SymbolicProcessing" → 0}];
Neftotal = 65.;
(*
Treh=1.0*10^(15);
gstar=100;
Nefpivot[VI_,Vend_,kpivot_]:=
 62.-Log[kpivot/(0.67/3000)]+Log[VI^(1/4)/10^(16)]+Log[VI^(1/4)/Vend^(1/4)]-
  (1/3.)*Log[(Vend^(1/4))*(Pi^2/30*gstar*Treh^(4))^(-1/4)];
*)
```

```
(*kpivot002=0.002;*)
kpivot05 = 0.05;
Clear[hend, VIt, kpivott, hpivot05, VI05];
tend1 = t /. FindRoot[ϵV[t] == 1., {t, tmin, tmax},
      AccuracyGoal → Automatic, PrecisionGoal → Automatic][[1]];
tend2 = t /. FindRoot[Abs[ηV[t]] == 1., {t, tmin, tmax},
      AccuracyGoal → Automatic, PrecisionGoal → Automatic][[1]];
tend = Max[tend1, tend2];
hend = hsol[t] /. t → tend;
(*VIt=VE[t]/.t→Log[MpR/mustart];*)
Vend = VE[t] /. t → tend;
kpivott = kpivot05;
tpivot05 = xxx /. Quiet[FindRoot[Nef[xxx, tend] == Neftotal, {xxx, tmin, tend}]];
hpivot05 = hsol[t] /. t → tpivot05;
VI05 = VE[tpivot05] / MpR^4;

tcrit2 = FindRoot[VEtt[t] / MpR^4 == 0, {t, tend, tpivot05}][[1, 2]];
hcrit = hsol[t] /. t → tcrit2;
```

Primordial black holes

```
efffactor = 0.4;
Msunkg = 2. × 10^30; (*in units of kg*)
kgtoGeV = 5.61 * 10^26;
Msun = Msunkg * kgtoGeV; (*in units of GeV*)
zetactest = 0.0875;
tEQ = 60. × 10^3; (*in units of years*)
aEQ = 3.7 × 10^-4;
Ommh2 = 0.12;
MEQ = 7 × 10^50 * 10^-3 / Msunkg;
M = 2.4 * 10^18; (*in units of GeV*)
MpRSUN = M / Msun; (*in units of Msun*)
Mstar = 3 * 10^12 * 10^-3 * kgtoGeV; (*in units of GeV*)
gstarFORM = 106.75;
hubble = 0.68;

betaFORM[t_, zetac_] := Erfc[zetac / Sqrt[2 Ps[t]]]
PBHmass[t_] := 4 π * efffactor * MpR^2 * Sqrt[3 MpR^2 / VE[t]] Exp[2 Nef[t, tend]]
PBHmassSUN[t_] := 4 π * efffactor * (MpR / Msun) * Sqrt[3 MpR^4 / VE[t]] Exp[2 Nef[t, tend]]
betaEQ[t_, zetac_] := betaFORM[t, zetac] aEQ / 1. Exp[Nef[tpivot05, tend] - Nef[t, tend]]
fraction[x_, zetac_] := 4.1 * 10^8 efffactor^(1/2) betaFORM[x, zetac] *
   (gstarFORM / 106.75)^(-1/4) (PBHmassSUN[x])^(-1/2) * (hubble / 0.68)^-2
```

```
findps[x_, zetac_] :=
 If[NumberQ[x] == True, FindRoot[1 == Erfc[zetac/Sqrt[2 PSvalue]] * Sqrt[MEQ/PBHmassSUN[x]], {PSvalue,
     1. × 10^-4, 1.}, AccuracyGoal → Automatic, PrecisionGoal → Automatic]][[1, 2]]
```

Result

Running of couplings

```
LogPlot[{g1sol[t], g2sol[t], g3sol[t], ytsol[t], lHsol[t]}, {t, 0, tmax},
 Frame → True, PlotRange → All, PlotStyle → {Black, Blue, Red, Green, Cyan}]
```




```
Plot[{0, lHsol[t], lHdsol[t]}, {t, 0, tmax}, Frame → True, PlotRange → All]
```



```
Plot[{xihsol[t]}, {t, 0, tmax}, Frame → True, PlotRange → All, PlotStyle → {Red, Blue},
 FrameLabel → {Style["t=ln(μ/Mt)", Medium], Style["Nonminimal couplings", Medium]}]
```




```
Plot[{sh[t] /. RGEsol}, {t, 0, tmax}, Frame → True, PlotRange → All, PlotStyle → {Blue}]
```



```
Plot[{GHintpln[t]}, {t, 0, tmax},
 Frame → True, PlotRange → All, PlotStyle → {Red, Blue},
 FrameLabel → {Style["t=ln(μ/Mt)", Medium], Style["G(t)", Medium]}]
```



Potential

```
ParametricPlot[{hsol[t] / MpR, VE[t] / MpR^4}, {t, 0, Log[0.05 * Mp / mustart]},
 PlotRange → {All, All}, Frame → True, AspectRatio → 1 / GoldenRatio,
 GridLines → {{{hcrit / MpR, {Dotted, Thick, Green}},
    {hpivot05 / MpR, Dashed}, {hend / MpR, Dashed}}, {}}]
```



Observables

```
hpivot05 / MpR
hend / MpR
hsol[tcrit] / MpR
hsol[tcrit2] / MpR
```

```
28.0955

0.333128

0.606643

0.41562
```

```
xihsol[tcrit2]
xihsol[tpivot05]
```

```
12.181

12.181
```

```
xihdsol[tcrit2]
xihdsol[tpivot05]
```

```
0.

0.
```

```
lHsol[tcrit2]
lHsol[tpivot05]
```

0.0000129305

0.00032094

```
lHdsol[tcrit2]
lHdsol[tpivot05]
```

-0.0000180473

0.000157606

```
lHddsol[tcrit2]
lHddsol[tpivot05]
```

0.0000491455

0.0000363375

```
Ps[t] /. t → tpivot05
ns[t] /. t → tpivot05
r[t] /. t → tpivot05
dnsdlnk[t] /. t → tpivot05
```

1.1875×10^{-7}

0.921084

0.294115

-0.00277184

```
Nef[tpivot05, tend]
```

NIntegrate::slwcon :
Numerical integration converging too slowly; suspect one of the following: singularity, value of the integration is 0, highly oscillatory integrand, or WorkingPrecision too small. ≫

NIntegrate::ncvb :
NIntegrate failed to converge to prescribed accuracy after 9 recursive bisections in x9 near {x9} = {35.0822}. NIntegrate obtained 65.00000000449137` and 53.54342620605956` for the integral and error estimates. ≫

65.

Slow-roll params

```
ParametricPlot[{hsol[t] / MpR, ϵV[t]}, {t, tend1, tpivot05}, PlotRange → All,
 Frame → True, AspectRatio → 1 / GoldenRatio, FrameLabel → {"h (in units of MP)", "ϵV"}]
```




```
ParametricPlot[{hsol[t] / MpR, Abs[ηV[t]]}, {t, tend2, tpivot05},
 PlotRange → All, Frame → True, AspectRatio → 1 / GoldenRatio,
 PlotRange → All, FrameLabel → {"h (in units of MP)", "|ηV|"}]
```




```
FindMinValue[ϵV[t], {t, tend, tpivot05}]
```

4.79964×10^{-13}

```
ParametricPlot[{hsol[t] / MpR, ϵV[t]}, {t, 34.8, 35.2}, PlotRange → All, Frame → True,
 AspectRatio → 1 / GoldenRatio, FrameLabel → {"h (in units of MP)", "ϵV"}]
```



Power spectrum & PBH

```
ParametricPlot[{hsol[t] / MpR, Nef[t, tend]},
 {t, tpivot05, tend}, AspectRatio → 1 / GoldenRatio, Frame → True,
 FrameLabel → {"h (in units of MP)", "N"}, PlotRange → All]
```




```
FindMaxValue[Log[10, Ps[t]], {t, tend, tpivot05}]
20.7702
```

```
ParametricPlot[{kpivot05 * Exp[Nef[tpivot05, tend] - Nef[t, tend]], Ps[t]},
 {t, tpivot05, tend}, AspectRatio → 1 / GoldenRatio, Frame → True, PlotRange → All,
 FrameLabel → {"k (in units of Mpc^-1)", "Pξ"}, ScalingFunctions → {"Log", "Log"},
 PlotRange → {{Log[10^-2], Log[10^26]}, {Log[10^-10], Log[10^-4]}}]
```

NIntegrate::slwcon :
Numerical integration converging too slowly; suspect one of the following: singularity, value of the integration is 0, highly oscillatory integrand, or WorkingPrecision too small. ≫

NIntegrate::ncvb :
NIntegrate failed to converge to prescribed accuracy after 9 recursive bisections in x9 near {x9} = {35.0822}. NIntegrate obtained 65.00000000449137` and 53.54342620605956` for the integral and error estimates. ≫

NIntegrate::slwcon :
Numerical integration converging too slowly; suspect one of the following: singularity, value of the integration is 0, highly oscillatory integrand, or WorkingPrecision too small. ≫

NIntegrate::ncvb :
NIntegrate failed to converge to prescribed accuracy after 9 recursive bisections in x9 near {x9} = {35.0822}. NIntegrate obtained 65.00000000449137` and 53.54342620605956` for the integral and error estimates. ≫